

A note concerning a BASIC display generator for search tasks

S. A. ROBERTSON

Manchester Polytechnic, Manchester, England M15 5FG

Stimulus materials for visual search experimentation of the sort performed by Neisser (1963) and later by Kristofferson (1972) often take the form of lists of, or matrices containing a number of background elements and a number of targets. In this context, elements and targets might well be numbers, letters, words, or other characters or character strings. In such experiments, an important variable underlying search or scanning time is the serial position of the target(s), that is, the number of earlier items (see Neisser, 1963).

A BASIC program is presented below which can conveniently prepare stimulus materials of this type. The user is able to determine the number of matrices and their size as well as the elements to be used as targets and background. These may be any characters or character strings. The program randomly generates a matrix of background elements and then allocates the target(s) to random positions within it. For each target, a count is made of the number of preceding elements, while a count is also made of the frequency of each nontarget; these details are output after each matrix.

Program Input. At Line 30 the number of matrices required (M) and the number of rows and columns required in each matrix (R and C) are input. Note that C is the number of elements in each row, where each element might occupy more than one character position as determined by the user. At Line 40 the number of targets (T) and the number of background elements (S) are input. Finally, the targets and then the background elements are input stepwise.

Table 1
A BASIC Display Generator for Search Tasks

```
10 REM SEARCH MATERIALS PROGRAM
20 DIM A$(50,50),T$(20),S$(100),T(20),S(100)
30 INPUT M,R,C
40 INPUT T,S
50 FOR I=0 TO T-1
60 INPUT T$(I)
70 NEXT I
80 FOR J=0 TO S-1
90 INPUT S$(J)
100 NEXT J
110 FOR U=1 TO M
120 MAT S = ZER
130 FOR K=0 TO R-1
140 FOR L=0 TO C-1
150 LET X=INT(S*RND(0))
160 LET A$(K,L)=S$(X)
170 LET S(X)=S(X)+1
180 NEXT L
190 NEXT K
```

```
200 FOR N=0 TO T-1
210 LET Y1=INT(R*RND(0))
220 LET Y2=INT(C*RND(0))
230 LET Z=0
240 IF A$(Y1,Y2)<>S$(Z) THEN 270
250 LET S(Z)=S(Z)-1
260 GOTO 300
270 LET Z=Z+1
280 IF Z=S THEN 210
290 GOTO 240
300 LET A$(Y1,Y2)=T$(N)
310 LET T(N)=Y1*C+Y2
320 NEXT N
330 FOR O=0 TO R-1
340 FOR P=0 TO C-1
350 PRINT A$(O,P);
360 NEXT P
370 PRINT
380 NEXT O
390 IF T=0 THEN 440
400 PRINT "TARGET", "NO. OF EARLIER ELEMENTS"
410 FOR Q=0 TO T-1
420 PRINT T$(Q),T(Q)
430 NEXT Q
440 PRINT "NONTARGETS", "FREQUENCY"
450 FOR Y=0 TO S-1
460 PRINT S$(Y),
470 IF S(Y)>=0 THEN 490
480 LET S(Y)=0
490 PRINT S(Y)
500 NEXT Y
510 NEXT U
520 STOP
530 END
```

Program Limitations. Apart from the DIM restrictions in Line 20, T may be any value ≥ 0 provided that $R*C \geq T$, as each target must be found a location in the matrix. S may be any value > 0 , subject to the same DIM statement. If the BASIC compiler used is unable to execute Line 120, this may be replaced by:

```
120 FOR A = 0 TO S - 1
122 S(A) = 0
124 NEXT A
```

Computer and Availability. The program was developed on an ICL 1902S computer. Sample output and program listings are available (at no cost) from the author at the School of Psychology, Manchester Polytechnic, Gaythorn Annexe, River Street, Manchester, England M15 5FG.

REFERENCES

- KRISTOFFERSON, M. W. Types and frequency of errors in visual search. *Perception & Psychophysics*, 1972, 11, 325-328.
- NEISSER, U. Decision-time without reaction-time: Experiments in visual scanning. *American Journal of Psychology*, 1963, 76, 376-385.

(Accepted for publication July 26, 1977.)